 Placement of holes

The following ideas revolve about the placement of holes and fitting of a plug with thin pins in a printed circuit board:

erase c1-c100

set c1

6(1)

end

set c2

1:6

end

plot c1*c2 c1*c2;           # This plot shows some holes and their pins.
noleg;

title 'Fitting of pins into holes';

tsize 1.1;

title 'Ring: the hole.     Filled circle: the pin.';

tsize 0.9;

tcolor 4;

overlay;

data 0 1 0 1;

symb;

type 1 6;

color 1 14;

size 10 6;

scale 1;

min 0;

max 7.

The main idea is that if the centers of the pins vary, what is then the probability that a plug with k pins will not fit the holes? This answer depends of course on the variation of the pins but also on the number of holes/pins in a plug. Here we suppose that the positions of the holes do not vary in any way. We also suppose that the variation in the X- and Y-directions can be described as normally distributed variables with mean zero and standard deviation sigma, i.e. N[0; sigma] (we avoid here the Greek alphabet) where sigma is equal in both X- and Y-directions.

The distance of the center of the pin from its target can be expressed as R = sqrt(X**2 + Y**2), i.e. We use the theorem of Pythagoras. The distribution of this radial distance R is called 'The Rayleigh distribution' (which in turn is a special case of the Weibull distribution, see below). The mean value is then 1.254 · sigma and the standard deviation is 0.655 · sigma where sigma is the value of the standard deviation of the X- and Y-deviations. This is derived using theoretical tools. (See also the document 'Rayleigh on the radio.doc' on our homepage www.ing-stat.se.)

Exercise.  You should now use the %Hdist-macro. After its initial run you change column c1 as per below. This macro treats the distance between a land (of copper) and a hole where both have variation. However, in our example with the plug, the holes have no variation (this assumption makes the problem easier). The mean value and the standard deviation of the land are accordingly set to zero in column c1 of the worksheet.

The mean of the hole displacement is zero but the standard deviation is 0.04. Change also the diameter of the hole to 0.30 mm. (The vertical line in the histogram is the difference (land diameter – hole diameter)/2.) After these changes you should rerun the macro:

      Land   Hole         Data %Hdist

      0.00   0.00  <-- Mean in mm (X)   

      0.00   0.00  <-- Mean in mm (Y)   

      0.00   0.04  <-- Std dev in mm (X)

      0.00   0.04  <-- Std dev in mm (Y)

      0.50   0.30  <-- Diameter         

   5000.00         <-- Number of values 

Results.  The macro describes the radial distance in column c10. Look in your session-window where you find the description of this and other columns. According to the theory of the Rayleigh distribution the mean value of c10 is expected to become 1.254*sigma i.e. 1.254*0.04 = 0.0501 and the standard deviation of 0.655*sigma i.e. 0.655*0.04 = 0.0262. In our simulation we got the values 0.0498 and 0.0263, i.e. rather close to the expected values which supports the theory. 
The Weibull distribution.  Before we continue with our example we will connect the Rayleigh to the Weibull distribution. The Weibull distribution is a well-known distribution derived by the Swedish scientist Weibull some 50 years ago. If you want to see its 'formula' you can look in either the Document ”A collection of diagrams” or via the menus [Calc]>[Probability Distributions]>[Weibull…] and then the HELP-button and the line Weibull distribution.

The formulas for the mean value and standard deviation of a Weibull distribution look (and are) very complicated (you can see them on the front page when you start the macro %WeibArea).

Now we have the following facts in this case:

· Sigma for the X- and Y-deviations is 0.04 (mm) according to our simulation input.

· Mean value of a Rayleigh distribution is 1.254*sigma, here 1.254*0.04 = 0.0501 (mm).

· The mean value of a Weibull distribution is stated as ”b*G(1 + 1/a)” where 'G' is
the so-called gamma function (not to be confused with the gamma distribution).

· The 'a'-value is 2 (see the first page of the %WeibArea-macro).

We use all this to get the 'b'-value for our particular distribution in this case. First we calculate the value of the gamma function when 'a' = 2 and then from this the 'b'-value (In Minitab 'a' and 'b' are called 'scale and shape parameters', respectively):

Let k1 = Gamma(1 + 1/2)            # Calc the gamma function at 1.5

Print k1                           # which is 0.886227.

# Using this we calculate 'b' as:

let k2 = 1.254*0.04/0.886227       # we use the expression for the mean

Print k2                           # to calculate b. (= 0.0565995).

Now we have the two parameters of a Weibull distribution (a = 2, and b = 0.0565995) which will give an area outside the red limit in the histogram from the simulation. In order to see this, you can use the menus [Calc]>[Probability Distributions]>[Weibull…] and then use the 'Cumulative Probability' and then enter the two parameter values in their fields. The limit 0.10 is entered in the 'Input constant'-field. The following commands do the same thing:

CDF 0.10;                          # 0.10 mm (the limit in the histogram).

Weibull 2 0.0565995.               # Gives the area left of 0.10 mm.

The area reported by Minitab is 0.9559 which is the area to the left of the red line in the histogram. Thus the area to the right is 1 – 0.9559 = 0.0441 which is rather close to the simulated result.

Conclusion.  From the reasoning above we draw the conclusion that if sigma is 0.04 mm and given the other assumptions we state that there is a 0.0441 (4 %) probability that a pin will not fit in the hole because its center is too far off. But in our initial drawing we have six holes. This means that we now need to use some probability reasoning again. In order for the plug to fit into the row of holes we have the following require​ments:

'The first hole must fit AND the second hole must fit AND the third hole must fit AND…'. This probability is then the first probability raised to the power of six:

let k3 = 0.9559**6                 # Or by using the [Calc]-menu. 

print k3                           # The result in the session window. 

This result becomes 0.762913 which gives 1 – 0.762913 = 0.237 as the probability that the 6-pin plug will not fit.

A final simulation.  We finish off this problem by a final simulation to confirm our results.
Erase c1-c100                      # Erases column c1 – c100.

Random 10000 c1-c12;               # X- and Y-deviations for 6 holes.

Normal 0 0.04.                     # Mean 0, stand dev 0.04.

                                   # Each row in the worksheet is now a

                                   # hole/pin-assembly.

Let c13 = sqrt(c1**2  + c2**2)     # Radial distance of hole 1.

Let c14 = sqrt(c3**2  + c4**2)     # Radial distance of hole 2.

Let c15 = sqrt(c5**2  + c6**2)     # Radial distance of hole 3.

Let c16 = sqrt(c7**2  + c8**2)     # Radial distance of hole 4.

Let c17 = sqrt(c9**2  + c10**2)    # Radial distance of hole 5.

Let c18 = sqrt(c11**2 + c12**2)    # Radial distance of hole 6.

# In order to be OK, all radial deviations must be within 0.10 mm.

# we use so-called logical function to calculate this: C20 will

# contain a '1' for every assembly that is OK:

Let c20 = (c13<0.10 and c14<0.10 and c15<0.10 and     &

           c16<0.10 and c17<0.10 and c18<0.10)

let k1 = 1 - sum(c20)/n(c20)       # Gives the proportion 'not OK'.

Prin k1

In the session-window we finally see an estimate of the proportion (probability) outside the speci​fication limit. This value is rather close to the theoretical value, 0.237, calculated above. In order to decrease this probability there are a few realistic and perhaps unrealistic things that can be done. Either increase the hole size or decrease the pin size or decrease the variation. Or perhaps a combi​nation of all these.

Another ambition.  If we are really ambitious we now perform these calculations (perhaps not the simulations) for a number of different sigma-values and perhaps sizes of plugs. Then we can draw good graphs of the result, perhaps using so-called regression analysis to see how the probability of result outside the limit changes. In this way we can get a material that can be used for discussions with designers, suppliers, quality engineers, etc, etc.

If you want to do this, you should extract all the necessary simulations or calculations into a text file. Then you should use constants (such as LET K20 = 0.10) in the beginning of your file and use 'k20' where applicable. In this way you have a full control over your program when changing values of the parameters. Perhaps you write a so-called DO-LOOP that conveniently runs through your logic and presents the results, etc. ■

